

Jun 06, 16 12:18

interpol.f90

Page 1/1

```

MODULE interpol
  USE iso_fortran_env, ONLY: WP => REAL64
  IMPLICIT NONE

CONTAINS

  SUBROUTINE polint(x,y,c)
    ! -----
    ! Calcula los coeficientes c(1),c(2),...c(n) del polinomio
    ! de grado n-1 que interpola los n puntos (x(1),y(1)),
    ! (x(2),y(2)),..., (x(n),y(n)) expresado según la fórmula
    ! de Newton:
    !
    !  $p(x) = c(1) + c(2)*(x-x(1)) + \dots + c(n)*(x-x(1))\dots(x-x(n-1))$ 
    ! -----
    REAL(WP), INTENT(IN) :: x(:)
    REAL(WP), INTENT(IN) :: y(:)
    REAL(WP), INTENT(OUT) :: c(:)
    ! -----
    INTEGER :: n,i,k
    ! -----
    n = SIZE(x)
    c(1) = y(1)
    DO k=2,n
      c(k) = y(k)
      DO i=1,k-1
        c(k) = (c(k)-c(i))/(x(k)-x(i))
      ENDDO
    ENDDO
  END SUBROUTINE polint

  REAL(WP) FUNCTION poleval(x,c,z)
    ! -----
    ! Evalua la forma de Newton del polinomio interpolante
    ! en un punto z dado.
    ! -----
    REAL(WP), INTENT(IN) :: x(:)
    REAL(WP), INTENT(IN) :: c(:)
    REAL(WP), INTENT(IN) :: z
    ! -----
    INTEGER :: n,k
    n = SIZE(x)
    poleval = c(n)
    DO k=n-1,1,-1
      poleval = c(k) + poleval*(z-x(k))
    ENDDO
  END FUNCTION poleval

END MODULE interpol

```